# Maintainer: Morten Linderud <foxboron@archlinux.org>
# Contributor: graysky <graysky AT archlinux dot us>

pkgname=lostfiles
pkgver=4.13
pkgrel=1
pkgdesc="Find orphaned files not owned by any Arch packages"
arch=(any)
url="https://github.com/graysky2/lostfiles"
license=(MIT)
backup=(etc/lostfiles.conf)
source=(${url}/archive/v${pkgver}/${pkgname}-${pkgver}.tar.gz)
sha256sums=('9df33b6fed1862301cd975e2dc607b2e82fe285b230a5469c16a5077460e9a4e')

build() {
  cd ${pkgname}-${pkgver}
  make 
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm0644 MIT "${pkgdir}/usr/share/licenses/${pkgname}/MIT"
}
